Rules for integrands of the form $(a + b \operatorname{Sin}[c + d x])^n$

1. $\int (b \operatorname{Sin}[c + d x])^n \, dx$

   1. $\int (b \operatorname{Sin}[c + d x])^n \, dx$ when $2 n \in \mathbb{Z}$

      1. $\int (b \operatorname{Sin}[c + d x])^n \, dx$ when $n > 1$

         1: $\int \operatorname{Sin}[c + d x]^n \, dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then $\operatorname{Sin}[c + d x]^n == -\frac{1}{d} \operatorname{Subst}\left[\left(1 - x^2\right)^{\frac{n-1}{2}}, x, \operatorname{Cos}[c + d x]\right] \partial_x \operatorname{Cos}[c + d x]$

Rule: If $\frac{n-1}{2} \in \mathbb{Z}^+$, then

$$\int \operatorname{Sin}[c + d x]^n \, dx \longrightarrow -\frac{1}{d} \operatorname{Subst}\left[\int \left(1 - x^2\right)^{\frac{n-1}{2}} dx, x, \operatorname{Cos}[c + d x]\right]$$

Program code:

```
Int[sin[c_.+d_.*x_]^n_,x_Symbol] :=
  -1/d*Subst[Int[Expand[(1-x^2)^((n-1)/2),x],x],x,Cos[c+d*x]] /;
FreeQ[{c,d},x] && IGtQ[(n-1)/2,0]
```

2. $\int (b\,\text{Sin}[c+d\,x])^n\,dx$ when $n>1$

1: $\int \text{Sin}[c+d\,x]^2\,dx$

Derivation: Algebraic expansion

Basis: $\sin[z]^2 == \frac{1}{2} - \frac{\cos[2z]}{2}$

Rule:

$$\int \text{Sin}[c+d\,x]^2\,dx \longrightarrow \frac{x}{2} - \frac{\text{Sin}[2c+2d\,x]}{4d}$$

Program code:

```
Int[sin[c_.+d_.*x_/2]^2,x_Symbol] :=
  x/2 - Sin[2*c+d*x]/(2*d) /;
FreeQ[{c,d},x]
```

2: $\int (b\,\text{Sin}[c+d\,x])^n\,dx$ when $n>1$

Reference: G&R 2.510.2 with $q \to 0$, CRC 299

Reference: G&R 2.510.5 with $p \to 0$, CRC 305

Derivation: Sine recurrence 3a with $A \to 0,\ B \to a,\ C \to b,\ m \to m-1,\ n \to -1$

Derivation: Sine recurrence 1b with $A \to 0,\ B \to 0,\ C \to b,\ a \to 0,\ m \to -1,\ n \to n-1$

Rule: If $n > 1$, then

$$\int (b \sin[c+d x])^n \, dx \longrightarrow -\frac{b \cos[c+d x] \, (b \sin[c+d x])^{n-1}}{d n} + \frac{b^2 (n-1)}{n} \int (b \sin[c+d x])^{n-2} \, dx$$

– Program code:

```
Int[(b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
(* -Cot[c+d*x]*(c*Sin[c+d*x])^n/(d*n) + b^2*(n-1)/n*Int[(b*Sin[c+d*x])^(n-2),x] *)
  -b*Cos[c+d*x]*(b*Sin[c+d*x])^(n-1)/(d*n) + b^2*(n-1)/n*Int[(b*Sin[c+d*x])^(n-2),x] /;
FreeQ[{b,c,d},x] && GtQ[n,1] && IntegerQ[2*n]
```

2: $\int (b \sin[c+d x])^n \, dx$ when $n < -1$

Reference: G&R 2.510.3 with $q \to 0$, CRC 309

Reference: G&R 2.510.6 with $p \to 0$, CRC 313

Reference: G&R 2.552.3

Derivation: Sine recurrence 3a with $A \to 0$, $B \to a$, $C \to b$, $m \to m-1$, $n \to -1$ inverted

– Derivation: Sine recurrence 2a with $A \to 1$, $B \to 0$, $C \to 0$, $a \to 0$, $m \to 0$

– Rule: If $n < -1$, then

$$\int (b \sin[c+d x])^n \, dx \longrightarrow \frac{\cos[c+d x] \, (b \sin[c+d x])^{n+1}}{b d (n+1)} + \frac{n+2}{b^2 (n+1)} \int (b \sin[c+d x])^{n+2} \, dx$$

– Program code:

```
Int[(b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Cos[c+d*x]*(b*Sin[c+d*x])^(n+1)/(b*d*(n+1)) +
  (n+2)/(b^2*(n+1))*Int[(b*Sin[c+d*x])^(n+2),x] /;
FreeQ[{b,c,d},x] && LtQ[n,-1] && IntegerQ[2*n]
```

3. $\int (b\,\text{Sin}[c+d\,x])^n\,dx$ when $-1 \le n \le -1$

   1. $\int \text{Sin}[c+d\,x]^n\,dx$ when $-1 \le n \le -1$

      1. $\int \text{Sin}[c+d\,x]^n\,dx$ when $n^2 == 1$

         1: $\int \text{Sin}[c+d\,x]\,dx$

Reference: G&R 2.01.5, CRC 290, A&S 4.3.113

Reference: G&R 2.01.6, CRC 291, A&S 4.3.114

Derivation: Primitive rule

Basis: $\partial_x \text{Cos}[c+d\,x] == -d\,\text{Sin}[c+d\,x]$

Rule:

$$\int \text{Sin}[c+d\,x]\,dx \longrightarrow -\frac{\text{Cos}[c+d\,x]}{d}$$

Program code:

```
Int[sin[c_.+Pi/2+d_.*x_],x_Symbol] :=
  Sin[c+d*x]/d /;
FreeQ[{c,d},x]
```

```
Int[sin[c_.+d_.*x_],x_Symbol] :=
  -Cos[c+d*x]/d /;
FreeQ[{c,d},x]
```

x: $\int \frac{1}{\text{Sin}[c + d\,x]}\,dx$

- Note: This rule not necessary since *Mathematica* automatically simplifies $\frac{1}{\text{Sin}[z]}$ to Csc[z].

- Rule:

$$\int \frac{1}{\text{Sin}[c + d\,x]}\,dx \longrightarrow \int \text{Csc}[c + d\,x]\,dx$$

- Program code:

```
(* Int[1/sin[c_.+d_.*x_],x_Symbol] :=
  Int[Csc[c+d*x],x] /;
FreeQ[{c,d},x] *)
```

2. $\int \text{Sin}[c + d\,x]^n\,dx$ when $n^2 == \frac{1}{4}$

1: $\int \sqrt{\text{Sin}[c + d\,x]}\,dx$

Derivation: Primitive rule

- Basis: $\partial_x\,\text{EllipticE}\left[\frac{1}{2}\left(x - \frac{\pi}{2}\right), 2\right] == \frac{\sqrt{\text{Sin}[x]}}{2}$

- Rule:

$$\int \sqrt{\text{Sin}[c + d\,x]}\,dx \longrightarrow \frac{2}{d}\,\text{EllipticE}\left[\frac{1}{2}\left(c - \frac{\pi}{2} + d\,x\right), 2\right]$$

- Program code:

```
Int[Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  2/d*EllipticE[1/2*(c-Pi/2+d*x),2] /;
FreeQ[{c,d},x]
```

2: $\int \frac{1}{\sqrt{\operatorname{Sin}[c+d\,x]}}\,dx$

Derivation: Primitive rule

Basis: $\partial_x \operatorname{EllipticF}\left[\frac{1}{2}\left(x-\frac{\pi}{2}\right),\,2\right] == \frac{1}{2\sqrt{\operatorname{Sin}[x]}}$

Rule:

$$\int \frac{1}{\sqrt{\operatorname{Sin}[c+d\,x]}}\,dx \longrightarrow \frac{2}{d}\operatorname{EllipticF}\left[\frac{1}{2}\left(c-\frac{\pi}{2}+d\,x\right),\,2\right]$$

Program code:

```
Int[1/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  2/d*EllipticF[1/2*(c-Pi/2+d*x),2] /;
FreeQ[{c,d},x]
```

2: $\int (b \operatorname{Sin}[c+d x])^n \, dx$ when $-1 < n < -1$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b \operatorname{Sin}[c+d x])^n}{\operatorname{Sin}[c+d x]^n} == 0$

Rule: If $-1 < n < -1$, then

$$\int (b \operatorname{Sin}[c+d x])^n \, dx \longrightarrow \frac{(b \operatorname{Sin}[c+d x])^n}{\operatorname{Sin}[c+d x]^n} \int \operatorname{Sin}[c+d x]^n \, dx$$

Program code:

```
Int[(b_*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Sin[c+d*x])^n/Sin[c+d*x]^n*Int[Sin[c+d*x]^n,x] /;
FreeQ[{b,c,d},x] && LtQ[-1,n,1] && IntegerQ[2*n]
```

2: $\int (b \operatorname{Sin}[c+d x])^n \, dx$ when $2 n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\operatorname{Cos}[c+d x]}{\sqrt{\operatorname{Cos}[c+d x]^2}} == 0$

Basis: $\frac{\operatorname{Cos}[c+d x]}{\sqrt{\operatorname{Cos}[c+d x]^2}} \frac{\operatorname{Cos}[c+d x]}{\sqrt{1-\operatorname{Sin}[c+d x]^2}} == 1$

Basis: $\operatorname{Cos}[c+d x] \, F[b \operatorname{Sin}[c+d x]] == \frac{1}{b d} \operatorname{Subst}[F[x], x, b \operatorname{Sin}[c+d x]] \, \partial_x (b \operatorname{Sin}[c+d x])$

Rule: If $2 n \notin \mathbb{Z}$, then

$$\int (b\,\text{Sin}[c+d\,x])^n\,dx \;\to\; \frac{\text{Cos}[c+d\,x]}{\sqrt{\text{Cos}[c+d\,x]^2}}\int \frac{\text{Cos}[c+d\,x]\,(b\,\text{Sin}[c+d\,x])^n}{\sqrt{1-\text{Sin}[c+d\,x]^2}}\,dx$$

$$\to\; \frac{\text{Cos}[c+d\,x]}{b\,d\,\sqrt{\text{Cos}[c+d\,x]^2}}\,\text{Subst}\Big[\int \frac{x^n}{\sqrt{1-\frac{x^2}{b^2}}}\,dx,\;x,\;b\,\text{Sin}[c+d\,x]\Big]$$

$$\to\; \frac{\text{Cos}[c+d\,x]\,(b\,\text{Sin}[c+d\,x])^{n+1}}{b\,d\,(n+1)\,\sqrt{\text{Cos}[c+d\,x]^2}}\,\text{Hypergeometric2F1}\Big[\frac{1}{2},\,\frac{n+1}{2},\,\frac{n+3}{2},\,\text{Sin}[c+d\,x]^2\Big]$$

Alternate rule: If $2\,n \notin \mathbb{Z}$, then

$$\int (b\,\text{Sin}[c+d\,x])^n\,dx \;\to\; -\frac{\text{Cos}[c+d\,x]\,(b\,\text{Sin}[c+d\,x])^{n+1}}{b\,d\,\left(\text{Sin}[c+d\,x]^2\right)^{\frac{n+1}{2}}}\,\text{Hypergeometric2F1}\Big[\frac{1}{2},\,\frac{1-n}{2},\,\frac{3}{2},\,\text{Cos}[c+d\,x]^2\Big]$$

Program code:

```
(* Int[(b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Cos[c+d*x]/(b*d*Sqrt[Cos[c+d*x]^2])*Subst[Int[x^n/Sqrt[1-x^2/b^2],x],x,b*Sin[c+d*x]] /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[2*n] || IntegerQ[3*n]] *)
```

```
Int[(b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Cos[c+d*x]*(b*Sin[c+d*x])^(n+1)/(b*d*(n+1)*Sqrt[Cos[c+d*x]^2])*Hypergeometric2F1[1/2,(n+1)/2,(n+3)/2,Sin[c+d*x]^2] /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[2*n]]
```

2: $\int (a+b\,\text{Sin}[c+d\,x])^2\,dx$

Derivation: Algebraic expansion

Basis: $(a+b\,z)^2 == \frac{1}{2}\left(2\,a^2+b^2\right)+2\,a\,b\,z-\frac{1}{2}\,b^2\left(1-2\,z^2\right)$

Rule:

$$\int (a + b \, \text{Sin}[c + d\, x])^2 \, dx \longrightarrow \frac{(2\, a^2 + b^2)\, x}{2} - \frac{2\, a\, b\, \text{Cos}[c + d\, x]}{d} - \frac{b^2\, \text{Cos}[c + d\, x]\, \text{Sin}[c + d\, x]}{2\, d}$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^2,x_Symbol] :=
  (2*a^2+b^2)*x/2 - 2*a*b*Cos[c+d*x]/d - b^2*Cos[c+d*x]*Sin[c+d*x]/(2*d) /;
FreeQ[{a,b,c,d},x]
```

3. $\int (a + b \, \text{Sin}[c + d\, x])^n \, dx$ when $a^2 - b^2 == 0$

 1. $\int (a + b \, \text{Sin}[c + d\, x])^n \, dx$ when $a^2 - b^2 == 0 \land 2\, n \in \mathbb{Z}$

  1. $\int (a + b \, \text{Sin}[c + d\, x])^n \, dx$ when $a^2 - b^2 == 0 \land 2\, n \in \mathbb{Z}^+$

   1: $\int (a + b \, \text{Sin}[c + d\, x])^n \, dx$ when $a^2 - b^2 == 0 \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2 - b^2 == 0 \land n \in \mathbb{Z}^+$, then

$$\int (a + b \, \text{Sin}[c + d\, x])^n \, dx \longrightarrow \int \text{ExpandTrig}\left[(a + b \, \text{Sin}[c + d\, x])^n,\, x\right] dx$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(a+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && IGtQ[n,0]
```

2. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land n + \frac{1}{2} \in \mathbb{Z}^+$

1: $\int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 == 0$

- Derivation: Singly degenerate sine recurrence 1b with $A \to c,\ B \to d,\ m \to \frac{1}{2},\ n \to -1,\ p \to 0$

- Rule: If $a^2 - b^2 == 0$, then

$$\int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow -\frac{2 b \operatorname{Cos}[c + d x]}{d \sqrt{a + b \operatorname{Sin}[c + d x]}}$$

- Program code:

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -2*b*Cos[c+d*x]/(d*Sqrt[a+b*Sin[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2-b^2==0\ \land\ n-\frac{1}{2}\in\mathbb{Z}^+$

- Reference: G&R 2.555.? inverted

- Derivation: Singly degenerate sine recurrence 1b with $A\to c,\ B\to d,\ n\to -1,\ p\to 0$

- Rule: If $a^2-b^2==0\ \land\ n-\frac{1}{2}\in\mathbb{Z}^+$, then

$$\int (a+b\,\text{Sin}[c+d\,x])^n\,dx\ \longrightarrow\ -\frac{b\,\text{Cos}[c+d\,x]\,(a+b\,\text{Sin}[c+d\,x])^{n-1}}{d\,n}+\frac{a\,(2\,n-1)}{n}\int (a+b\,\text{Sin}[c+d\,x])^{n-1}\,dx$$

- Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n-1)/(d*n) +
  a*(2*n-1)/n*Int[(a+b*Sin[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && IGtQ[n-1/2,0]
```

2. $\int (a + b \sin[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2 n \in \mathbb{Z}^-$

1: $\int \frac{1}{a + b \sin[c + d x]} \, dx$ when $a^2 - b^2 == 0$

Reference: G&R 2.555.3', CRC 337', A&S 4.3.134'/5'

Derivation: Singly degenerate sine recurrence 2a with $A \to 1$, $B \to 0$, $m \to -1$, $n \to 0$, $p \to 0$

Rule: If $a^2 - b^2 == 0$, then

$$\int \frac{1}{a + b \sin[c + d x]} \, dx \longrightarrow -\frac{\cos[c + d x]}{d \, (b + a \sin[c + d x])}$$

Program code:

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -Cos[c+d*x]/(d*(b+a*Sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

2: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$ when $a^2 - b^2 == 0$

Derivation: Integration by substitution

Basis: If $a^2 - b^2 == 0$, then $\frac{1}{\sqrt{a+b \operatorname{Sin}[c+d x]}} == -\frac{2}{d} \operatorname{Subst}\left[\frac{1}{2 a-x^2}, x, \frac{b \operatorname{Cos}[c+d x]}{\sqrt{a+b \operatorname{Sin}[c+d x]}}\right] \partial_x \frac{b \operatorname{Cos}[c+d x]}{\sqrt{a+b \operatorname{Sin}[c+d x]}}$

Rule: If $a^2 - b^2 == 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow -\frac{2}{d} \operatorname{Subst}\left[\int \frac{1}{2 a - x^2} \, dx, x, \frac{b \operatorname{Cos}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]}}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -2/d*Subst[Int[1/(2*a-x^2),x],x,b*Cos[c+d*x]/Sqrt[a+b*Sin[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

3: $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land n < -1 \land 2 n \in \mathbb{Z}$

Reference: G&R 2.555.?

Derivation: Singly degenerate sine recurrence 2a with $A \to 1,\ B \to 0,\ n \to 0,\ p \to 0$

Rule: If $a^2 - b^2 == 0 \land n < -1 \land 2 n \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sin}[c + d x])^n \, dx \to \frac{b \operatorname{Cos}[c + d x] \, (a + b \operatorname{Sin}[c + d x])^n}{a d \, (2 n + 1)} + \frac{n + 1}{a \, (2 n + 1)} \int (a + b \operatorname{Sin}[c + d x])^{n+1} \, dx$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  b*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(a*d*(2*n+1)) +
  (n+1)/(a*(2*n+1))*Int[(a+b*Sin[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

2. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2 n \notin \mathbb{Z}$

x: $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2 n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2 - b^2 == 0$, then $\partial_x \frac{\operatorname{Cos}[c+d x]}{\sqrt{a-b \operatorname{Sin}[c+d x]} \, \sqrt{a+b \operatorname{Sin}[c+d x]}} == 0$

Basis: If $a^2 - b^2 == 0$, then $\frac{a^2 \operatorname{Cos}[c+d x]}{\sqrt{a+b \operatorname{Sin}[c+d x]} \, \sqrt{a-b \operatorname{Sin}[c+d x]}} \frac{\operatorname{Cos}[c+d x]}{\sqrt{a+b \operatorname{Sin}[c+d x]} \, \sqrt{a-b \operatorname{Sin}[c+d x]}} == 1$

Basis: $\operatorname{Cos}[c + d x] \, F[\operatorname{Sin}[c + d x]] == \frac{1}{d} \operatorname{Subst}[F[x], x, \operatorname{Sin}[c + d x]] \, \partial_x \operatorname{Sin}[c + d x]$

Note: If $3 n \in \mathbb{Z}$, this results in a complicated expression involving elliptic integrals instead of a single hypergeometric function.

- Rule: If $a^2-b^2==0 \land 2n\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sin}[c+d\,x])^n\,dx \longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[c+d\,x]}{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}\,\sqrt{a-b\,\mathrm{Sin}[c+d\,x]}}\int\frac{\mathrm{Cos}[c+d\,x]\,(a+b\,\mathrm{Sin}[c+d\,x])^{n-\frac{1}{2}}}{\sqrt{a-b\,\mathrm{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[c+d\,x]}{d\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}\,\sqrt{a-b\,\mathrm{Sin}[c+d\,x]}}\,\mathrm{Subst}\left[\int\frac{(a+b\,x)^{n-\frac{1}{2}}}{\sqrt{a-b\,x}}\,dx,\,x,\,\mathrm{Sin}[c+d\,x]\right]$$

- Program code:

```
(* Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^2*Cos[c+d*x]/(d*Sqrt[a+b*Sin[c+d*x]]*Sqrt[a-b*Sin[c+d*x]])*Subst[Int[(a+b*x)^(n-1/2)/Sqrt[a-b*x],x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] *)
```

1: $\int (a+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2-b^2==0\land 2n\notin\mathbb{Z}\land a>0$

- Derivation: Piecewise constant extraction and integration by substitution

- Rule: If $a^2-b^2==0\land 2\,n\notin\mathbb{Z}\land a>0$, then

$$\int (a+b\,\text{Sin}[c+d\,x])^n\,dx\longrightarrow$$

$$-\frac{2^{n+\frac{1}{2}}\,a^{n-\frac{1}{2}}\,b\,\text{Cos}[c+d\,x]}{d\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,\text{Hypergeometric2F1}\left[\frac{1}{2},\frac{1}{2}-n,\frac{3}{2},\frac{1}{2}\left(1-\frac{b\,\text{Sin}[c+d\,x]}{a}\right)\right]$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -2^(n+1/2)*a^(n-1/2)*b*Cos[c+d*x]/(d*Sqrt[a+b*Sin[c+d*x]])*Hypergeometric2F1[1/2,1/2-n,3/2,1/2*(1-b*Sin[c+d*x]/a)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && GtQ[a,0]
```

2: $\int (a + b \,\mathrm{Sin}[c + d\,x])^n \,dx$ when $a^2 - b^2 == 0 \land 2n \notin \mathbb{Z} \land a \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b\,\mathrm{Sin}[c+d\,x])^n}{\left(1+\frac{b}{a}\,\mathrm{Sin}[c+d\,x]\right)^n} == 0$

Rule: If $a^2 - b^2 == 0 \land 2n \notin \mathbb{Z} \land a \ngtr 0$, then

$$\int (a + b\,\mathrm{Sin}[c + d\,x])^n \,dx \longrightarrow \frac{a^{\mathrm{IntPart}[n]}\,(a + b\,\mathrm{Sin}[c + d\,x])^{\mathrm{FracPart}[n]}}{\left(1 + \frac{b}{a}\,\mathrm{Sin}[c + d\,x]\right)^{\mathrm{FracPart}[n]}} \int \left(1 + \frac{b}{a}\,\mathrm{Sin}[c + d\,x]\right)^n dx$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^IntPart[n]*(a+b*Sin[c+d*x])^FracPart[n]/(1+b/a*Sin[c+d*x])^FracPart[n]*Int[(1+b/a*Sin[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && Not[GtQ[a,0]]
```

4. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0$

   1. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}$

      1. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}^+$

         1. $\int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 \neq 0$

            1: $\int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 \neq 0 \land a + b > 0$

Derivation: Primitive rule

Basis: If $a + b > 0$, then $\partial_x \operatorname{EllipticE}\left[\frac{1}{2}\left(x - \frac{\pi}{2}\right), \frac{2 b}{a+b}\right] == \frac{1}{2\sqrt{a+b}} \sqrt{a + b \operatorname{Sin}[x]}$

Rule: If $a^2 - b^2 \neq 0 \land a + b > 0$, then

$$\int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx \to \frac{2\sqrt{a + b}}{d} \operatorname{EllipticE}\left[\frac{1}{2}\left(c - \frac{\pi}{2} + d x\right), \frac{2 b}{a + b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2*Sqrt[a+b]/d*EllipticE[1/2*(c-Pi/2+d*x),2*b/(a+b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a+b,0]
```

2: $\int \sqrt{a + b \operatorname{Sin}[c + d\,x]}\, dx$ when $a^2 - b^2 \neq 0 \land a - b > 0$

Derivation: Primitive rule

Basis: If $a - b > 0$, then $\partial_x \operatorname{EllipticE}\left[\frac{1}{2}\left(x + \frac{\pi}{2}\right), -\frac{2\,b}{a-b}\right] == \frac{1}{2\sqrt{a-b}} \sqrt{a + b \operatorname{Sin}[x]}$

Rule: If $a^2 - b^2 \neq 0 \land a - b > 0$, then

$$\int \sqrt{a + b \operatorname{Sin}[c + d\,x]}\, dx \to \frac{2\sqrt{a-b}}{d} \operatorname{EllipticE}\left[\frac{1}{2}\left(c + \frac{\pi}{2} + d\,x\right), -\frac{2\,b}{a-b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2*Sqrt[a-b]/d*EllipticE[1/2*(c+Pi/2+d*x),-2*b/(a-b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a-b,0]
```

3: $\int \sqrt{a + b \operatorname{Sin}[c + d\,x]}\, dx$ when $a^2 - b^2 \neq 0 \land a + b \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b\,f[x]}}{\sqrt{\frac{a+b\,f[x]}{a+b}}} == 0$

Note: Since $\frac{a}{a+b} + \frac{b}{a+b} == 1 > 0$, the above rule applies to the resulting integrand.

Rule: If $a^2 - b^2 \neq 0 \land a + b \ngtr 0$, then

$$\int \sqrt{a + b \, \text{Sin}[c + d \, x]} \, dx \longrightarrow \frac{\sqrt{a + b \, \text{Sin}[c + d \, x]}}{\sqrt{\frac{a+b \, \text{Sin}[c+d \, x]}{a+b}}} \int \sqrt{\frac{a}{a + b} + \frac{b}{a + b} \, \text{Sin}[c + d \, x]} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[a+b*Sin[c+d*x]]/Sqrt[(a+b*Sin[c+d*x])/(a+b)]*Int[Sqrt[a/(a+b)+b/(a+b)*Sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && Not[GtQ[a+b,0]]
```

2: $\int (a + b \, \text{Sin}[c + d \, x])^n \, dx$ when $a^2 - b^2 \neq 0 \land n > 1 \land 2 n \in \mathbb{Z}$

– Derivation: Nondegenerate sine recurrence 1b with $A \to a \, c$, $B \to b \, c + a \, d$, $C \to b \, d$, $m \to -1 + m$, $n \to -1$, $p \to 0$

– Rule: If $a^2 - b^2 \neq 0 \land n > 1 \land 2 n \in \mathbb{Z}$, then

$$\int (a + b \, \text{Sin}[c + d \, x])^n \, dx \longrightarrow$$
$$-\frac{b \, \text{Cos}[c + d \, x] \, (a + b \, \text{Sin}[c + d \, x])^{n-1}}{d \, n} + \frac{1}{n} \int (a + b \, \text{Sin}[c + d \, x])^{n-2} \left(a^2 \, n + b^2 \, (n - 1) + a \, b \, (2 \, n - 1) \, \text{Sin}[c + d \, x]\right) dx$$

– Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n-1)/(d*n) +
  1/n*Int[(a+b*Sin[c+d*x])^(n-2)*Simp[a^2*n+b^2*(n-1)+a*b*(2*n-1)*Sin[c+d*x],x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

2. $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}^-$

   1. $\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 \neq 0$

      1. $\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 > 0$

         1: $\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 > 0 \land a > 0$

–

Note: Resulting antiderivative is continuous on the real line.

Rule: If $a^2 - b^2 > 0 \land a > 0$, let $q = \sqrt{a^2 - b^2}$, then

$$\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow \frac{x}{q} + \frac{2}{d q} \operatorname{ArcTan}\left[\frac{b \operatorname{Cos}[c + d x]}{a + q + b \operatorname{Sin}[c + d x]}\right]$$

Program code:

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  With[{q=Rt[a^2-b^2,2]},
  x/q + 2/(d*q)*ArcTan[b*Cos[c+d*x]/(a+q+b*Sin[c+d*x])]] /;
FreeQ[{a,b,c,d},x] && GtQ[a^2-b^2,0] && PosQ[a]
```

2: $\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx$ when $a^2 - b^2 > 0 \land a \ngtr 0$

- Note: Resulting antiderivative is continuous on the real line.

- Rule: If $a^2 - b^2 > 0 \land a \ngtr 0$, let $q = \sqrt{a^2 - b^2}$, then

$$\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow -\frac{x}{q} - \frac{2}{d q} \operatorname{ArcTan}\left[\frac{b \operatorname{Cos}[c + d x]}{a - q + b \operatorname{Sin}[c + d x]}\right]$$

- Program code:

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  With[{q=Rt[a^2-b^2,2]},
  -x/q - 2/(d*q)*ArcTan[b*Cos[c+d*x]/(a-q+b*Sin[c+d*x])]] /;
FreeQ[{a,b,c,d},x] && GtQ[a^2-b^2,0] && NegQ[a]
```

2: $\int \frac{1}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$ when $a^2-b^2\neq 0$

Reference: G&R 2.551.3, CRC 340, A&S 4.3.131

Reference: G&R 2.553.3, CRC 341, A&S 4.3.133

Derivation: Integration by substitution

Basis:
$F[\mathrm{Sin}[c+d\,x],\,\mathrm{Cos}[c+d\,x]] == \frac{2}{d}\,\mathrm{Subst}\left[\frac{1}{1+x^2}\,F\left[\frac{2\,x}{1+x^2},\,\frac{1-x^2}{1+x^2}\right],\,x,\,\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]\right]\,\partial_x\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]$

Basis: $\frac{1}{a+b\,\mathrm{Sin}[c+d\,x]} == \frac{2}{d}\,\mathrm{Subst}\left[\frac{1}{a+2\,b\,x+a\,x^2},\,x,\,\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]\right]\,\partial_x\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]$

Basis: $\frac{1}{a+b\,\mathrm{Cos}[c+d\,x]} == \frac{2}{d}\,\mathrm{Subst}\left[\frac{1}{a+b+(a-b)\,x^2},\,x,\,\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]\right]\,\partial_x\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]$

Note: $\mathrm{Tan}\left[\frac{z}{2}\right] == \frac{\mathrm{Sin}[z]}{1+\mathrm{Cos}[z]}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{1}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx \longrightarrow \frac{2}{d}\,\mathrm{Subst}\left[\int \frac{1}{a+2\,b\,x+a\,x^2}\,dx,\,x,\,\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]\right]$$

$$\int \frac{1}{a+b\,\mathrm{Cos}[c+d\,x]}\,dx \longrightarrow \frac{2}{d}\,\mathrm{Subst}\left[\int \frac{1}{a+b+(a-b)\,x^2}\,dx,\,x,\,\mathrm{Tan}\left[\frac{1}{2}\,(c+d\,x)\right]\right]$$

Program code:

```
Int[1/(a_+b_.*sin[c_.+Pi/2+d_.*x_]),x_Symbol] :=
  With[{e=FreeFactors[Tan[(c+d*x)/2],x]},
  2*e/d*Subst[Int[1/(a+b+(a-b)*e^2*x^2),x],x,Tan[(c+d*x)/2]/e]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  With[{e=FreeFactors[Tan[(c+d*x)/2],x]},
  2*e/d*Subst[Int[1/(a+2*b*e*x+a*e^2*x^2),x],x,Tan[(c+d*x)/2]/e]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

2. $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$ when $a^2 - b^2 \neq 0$

1: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$ when $a^2 - b^2 \neq 0 \land a + b > 0$

Derivation: Primitive rule

Basis: If $a + b > 0$, then $\partial_x \operatorname{EllipticF}\left[\frac{1}{2} \left(x - \frac{\pi}{2}\right), \frac{2 b}{a+b}\right] == \frac{\sqrt{a+b}}{2 \sqrt{a+b \operatorname{Sin}[x]}}$

Rule: If $a^2 - b^2 \neq 0 \land a + b > 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \to \frac{2}{d \sqrt{a + b}} \operatorname{EllipticF}\left[\frac{1}{2} \left(c - \frac{\pi}{2} + d x\right), \frac{2 b}{a + b}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a+b])*EllipticF[1/2*(c-Pi/2+d*x),2*b/(a+b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a+b,0]
```

2: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$ when $a^2 - b^2 \neq 0 \land a - b > 0$

Derivation: Primitive rule

Basis: If $a - b > 0$, then $\partial_x \operatorname{EllipticF}\left[\frac{1}{2}\left(x + \frac{\pi}{2}\right), -\frac{2 b}{a-b}\right] == \frac{\sqrt{a-b}}{2 \sqrt{a+b \operatorname{Sin}[x]}}$

Rule: If $a^2 - b^2 \neq 0 \land a - b > 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \rightarrow \frac{2}{d \sqrt{a - b}} \operatorname{EllipticF}\left[\frac{1}{2}\left(c + \frac{\pi}{2} + d x\right), -\frac{2 b}{a - b}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a-b])*EllipticF[1/2*(c+Pi/2+d*x),-2*b/(a-b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a-b,0]
```

3: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$ when $a^2 - b^2 \neq 0 \land a + b \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\frac{a+b f[x]}{a+b}}}{\sqrt{a+b f[x]}} == 0$

Note: Since $\frac{a}{a+b} + \frac{b}{a+b} == 1 > 0$, rule f1 applies to the resulting integrand.

Rule: If $a^2 - b^2 \neq 0 \land a + b \ngtr 0$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx \longrightarrow \frac{\sqrt{\frac{a+b\,\text{Sin}[c+d\,x]}{a+b}}}{\sqrt{a+b\,\text{Sin}[c+d\,x]}}\int \frac{1}{\sqrt{\frac{a}{a+b}+\frac{b}{a+b}\,\text{Sin}[c+d\,x]}}\,dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[(a+b*Sin[c+d*x])/(a+b)]/Sqrt[a+b*Sin[c+d*x]]*Int[1/Sqrt[a/(a+b)+b/(a+b)*Sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && Not[GtQ[a+b,0]]
```

3: $\int (a+b\,\text{Sin}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\land n<-1\land 2n\in\mathbb{Z}$

Reference: G&R 2.552.3

Derivation: Nondegenerate sine recurrence 1a with $A\to 1,\ B\to 0,\ C\to 0,\ m\to 0,\ p\to 0$

Rule: If $a^2-b^2\neq 0\land n<-1\land 2n\in\mathbb{Z}$, then

$$\int (a+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow$$
$$-\frac{b\,\text{Cos}[c+d\,x]\,(a+b\,\text{Sin}[c+d\,x])^{n+1}}{d\,(n+1)\,(a^2-b^2)}+\frac{1}{(n+1)\,(a^2-b^2)}\int (a+b\,\text{Sin}[c+d\,x])^{n+1}\,(a\,(n+1)-b\,(n+2)\,\text{Sin}[c+d\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  1/((n+1)*(a^2-b^2))*Int[(a+b*Sin[c+d*x])^(n+1)*Simp[a*(n+1)-b*(n+2)*Sin[c+d*x],x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

2: $\int (a + b \operatorname{Sin}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\operatorname{Cos}[c+d x]}{\sqrt{1+\operatorname{Sin}[c+d x]} \sqrt{1-\operatorname{Sin}[c+d x]}} == 0$

Basis: $\operatorname{Cos}[c + d x] == \frac{1}{d} \partial_x \operatorname{Sin}[c + d x]$

Rule: If $a^2 - b^2 \neq 0 \land 2 n \notin \mathbb{Z}$, then

$$\int (a + b \operatorname{Sin}[c + d x])^n \, dx \longrightarrow$$

$$\frac{\operatorname{Cos}[c + d x]}{\sqrt{1 + \operatorname{Sin}[c + d x]} \sqrt{1 - \operatorname{Sin}[c + d x]}} \int \frac{\operatorname{Cos}[c + d x] (a + b \operatorname{Sin}[c + d x])^n}{\sqrt{1 + \operatorname{Sin}[c + d x]} \sqrt{1 - \operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$

$$\frac{\operatorname{Cos}[c + d x]}{d \sqrt{1 + \operatorname{Sin}[c + d x]} \sqrt{1 - \operatorname{Sin}[c + d x]}} \operatorname{Subst}\left[\int \frac{(a + b x)^n}{\sqrt{1 + x} \sqrt{1 - x}} \, dx, x, \operatorname{Sin}[c + d x]\right]$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Cos[c+d*x]/(d*Sqrt[1+Sin[c+d*x]]*Sqrt[1-Sin[c+d*x]])*Subst[Int[(a+b*x)^n/(Sqrt[1+x]*Sqrt[1-x]),x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*n]]
```

Rules for integrands of the form $(a + b \operatorname{Sin}[c + d x] \operatorname{Cos}[c + d x])^n$

1: $\int (a + b \operatorname{Sin}[c + d x] \operatorname{Cos}[c + d x])^n \, dx$

Derivation: Algebraic simplification

Basis: $\operatorname{Sin}[z] \operatorname{Cos}[z] == \frac{1}{2} \operatorname{Sin}[2 z]$

Rule:

$$\int (a + b \operatorname{Sin}[c + d x] \operatorname{Cos}[c + d x])^n \, dx \longrightarrow \int \left(a + \frac{1}{2} b \operatorname{Sin}[2 c + 2 d x]\right)^n dx$$

Program code:

```
Int[(a_+b_.*sin[c_.+d_.*x_]*cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Sin[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d,n},x]
```